数値計算のスクエア
テイラー · オイラーとルンゲ · クッター
(計算とスクリプト)

Shimura Masato
JCD02773@nifty.ne.jp

2009 年 2 月 25 日

目次

1 テーラー級数

1.1 テーラー展開の技

A 石村/石村にテーラー展開の技として次の例が紹介されている。

$f(x) = 7x^4 - 11x^3 + 5x^2 - 3x + 8$

J の	多項式の定義用に形式を合わせる。			
$f(x) =$	$8 - 3x + 5x^2 - 11x^3 + 7x^4$	$\rightarrow$	$f(0) =$	8
$f'(x) =$	$-3 + 5 \cdot 2x - 11 \cdot 3x^2 + 7 \cdot 4x^3$	$\rightarrow$	$f'(0) =$	-3
$f''(x) =$	$5 \cdot 2 - 11 \cdot 3 \cdot 2x + 7 \cdot 4 \cdot 3x^2$	$\rightarrow$	$f''(0) =$	$5 \cdot 2!$
$f'''(x) =$	$-11 \cdot 3 \cdot 2x + 7 \cdot 4 \cdot 3 \cdot 2x$	$\rightarrow$	$f'''(0) =$	$-11 \cdot 3!$
$f^{(4)}(x) =$	$7 \cdot 4 \cdot 3 \cdot 2$	$\rightarrow$	$f^{(4)}(0) =$	$7 \cdot 4!$
$7^{(5} =$	0	$\rightarrow$	$f^{(5)}(0) =$	0

$\to -3 = \dfrac{f'(0)}{1!}$
$\to 5 = \dfrac{f''(0)}{2!}$
$\to -11 = \dfrac{f'''(0)}{3!}$
$\to -7 = \dfrac{f''''(0)}{4!}$

$$f(x) = 8 - 3x + 5x^2 - 11x^3 + 7x^4 = f(0) + \frac{f'(0)}{1!}x + \frac{f''(0)}{2!}x^2 + \frac{f'''(0)}{3!}x^3 + \frac{f''''(0)}{3!}x^4$$

Z　多項式で回帰するときに骨格は低次の項で固まり、高次の項は細かい波形で微調整の役割を果たすが、ここでも低次の項が先に顔を出し、高次の項が誤差の調整をしている。

A　運動会では数字の少ない方がよいし、律令時代の官位も小さい方が上だ。

1.2　e^x の微分と計算

$e^x = 1 + x + \dfrac{x^2}{2!} + \dfrac{x^3}{3!} + \cdots + \dfrac{x^n}{n!}$
とする。$x = 1$ の場合は

$e_0 =$　1
$e_1 =$　2
$e_2 =$　2.5
$e_3 =$　2.66667
$e_4 =$　2.70833
$e_5 =$　2.71667

```
] a=. |: L:0 ({@> >:i.6) ex0 (L:0) 1
+---+---+---+---+----+-----+
|1 1|1 1|1 1|1 1|1  1|1   1|
|   |1 1|1 1|1 1|1  1|1   1|
|   |   |1 2|1 2|1  2|1   2|
|   |   |   |1 6|1  6|1   6|
|   |   |   |   |1 24|1  24|
|   |   |   |   |    |1 120|
+---+---+---+---+----+-----+

 (i. L:0 # L:0 tmp),.  L:0 tmp=. %/"1 L:0 a
+---+---+-----+---------+----------+-----------+
|0 1|0 1|0   1|0        1|0          1|0           1|
```

```
|   |1 1|1   1|1         1|1          1|1           1|
|   |   |2 0.5|2       0.5|2        0.5|2         0.5|
|   |   |     |3 0.166667|3   0.166667|3    0.166667|
|   |   |     |          |4 0.0416667|4   0.0416667|
|   |   |     |          |           |5  0.00833333|
+---+---+-----+----------+-----------+-------------+

 +/ L:0  %/"1 L:0 a

+-+-+---+-------+------+-------+
|1|2|2.5|2.66667|2.70833|2.71667|
+-+-+---+-------+------+-------+
```

1.3　$cosx$ の項毎の微分

$$cosx = 1 - \frac{1}{2!}x^2 + \frac{1}{4!}x^4 - \frac{1}{6!}x^6 + \frac{1}{8!}x^8$$

$$\frac{d}{dx}cosx = 0 - \frac{2}{2!}x^{2-1} + \frac{4}{4!}x^{4-1} - \frac{6}{6!}x^{6-1} + \frac{8}{8!}x^{8-1}$$
$$= -(x - \frac{1}{3!}x^3 + \frac{1}{5!}x^5 - \frac{1}{7!}x^7 \cdots) = -sinx$$

```
   1&o. t. i.6  NB.1&o. is cos
0 1 0 _0.166667 0 0.00833333

  +/ 1&o. t. i.6 NB. sum
0.841667
```

1.4　Script

```
NB. calc exp taylor value
NB. Usage: 10 ex0 5/10 ex 5
ex0=:(] ^ i.@[) ,: !&i.@[
ex=:+/@((] ^ i.@[) %/"0 !&i.@[)
```

2　オイラー法とルンゲ · クッタ法

西川「J による微分方程式のグラフィック · アプローチ（その 1）」(JAPLA 2007) は数式とおり入力し、正規表現で J の関数に変換し、解のグラフとベクトル場を表示する本格的なものである。この中にオイラー法からルンゲ · クッタ法までのスクリプトが入っている。この 4 本の Script を取り出し多少修正した。

2.1　4 本の Script

関数を左引数とする副詞型

method	*name*
Euler	*euler*0
ModifiedEuler	*eulerx*
Heun	*heun*
Runge · Kutter	*rk*

Δx	左引数	x
初期値	右引数	y
関数	個別に定義	u

EX. $\dfrac{dx}{dy} = -2 + 3e^x$

`f3=: 3 : ' (3 * 1x1 ^ {.y)- ^&2 {:y'` NB. Hattori Ex4.1

```
0 1 3 5 7{"1
   (0.02 f3 euler0 ^:(i.20) 0 1 ),.
   (0.02 f3 eulerx ^:(i.20) 0 1),.
   (0.02 f3 heun ^:(i.20)  0 1),.
    0.02 f3 rk ^:(i.20) 0  1

 t(i)  euler0 eulerx  heun  runge.kutter
-------------------------------------------
   0        1        1        1        1
0.02    1.04  1.0398 1.03979  1.0398
0.04 1.07958 1.07917 1.07916 1.07917
```

```
0.06 1.11872  1.1181 1.11809  1.1181
0.08  1.1574 1.15657 1.15656 1.15658
 0.1  1.1956 1.19458 1.19456 1.19459
0.12 1.23333 1.23211 1.23208 1.23212
0.14 1.27055 1.26915 1.26912 1.26916
0.16 1.30728  1.3057 1.30567 1.30572
0.18 1.34351 1.34177 1.34174 1.34179
 0.2 1.37925 1.37735 1.37732 1.37737
0.22 1.41448 1.41245 1.41242 1.41247
0.24 1.44923 1.44707 1.44704  1.4471
0.26  1.4835 1.48123  1.4812 1.48126
0.28  1.5173 1.51494 1.51491 1.51497
 0.3 1.55065  1.5482 1.54817 1.54824
0.32 1.58355 1.58104 1.58101 1.58108
0.34 1.61602 1.61347 1.61344  1.6135
0.36 1.64809  1.6455 1.64547 1.64554
0.38 1.67977 1.67716 1.67713  1.6772
```

2.2　2 階微分方程式

2 階微分方程式： $\frac{d^2y}{dx^2} + 4\frac{dy}{dx} + 4y = 0$

1 階連立微分方程式に書き直す。

$$\begin{cases} \frac{dy}{dx} = z \\ \frac{dz}{dx} = -4(y + z) \end{cases}$$

初期値 $x = 0, y = 1, \frac{dy}{dx} = 0(= z), h = 0.2$

連立方程式の関数定義。; で 2 の関数を繋ぎ L:0 で同時に計算する。

```
f10=:3 : '({: y); _4* +/ }. y'

 0.2 f10 rk_2nd ^:(i.6) 0 1 0  NB. repeat 6 times
x        y         z
--------------------
0        1         0
0.2 0.938133 _0.535467
```

```
0.4 0.808413 _0.717954
0.6 0.662289 _0.721974
0.8 0.524667 _0.645349
  1 0.405817 _0.540802
```

2 階微分方程式 : $\dfrac{d^2y}{dx^2} + \dfrac{dy}{dx} + 8y = 0$

1 階連立微分方程式に書き直す。

$$\begin{cases} \dfrac{dy}{dx} = z \\ \dfrac{dz}{dx} = -8y - z \end{cases}$$

初期値 $x = 0, y = 0.5, \dfrac{dy}{dx} = 1(= z), h = 0.1$

```
plot {2{. |:  0.1 f11 rk_2nd ^:(i.50) 0 0.5 1
```



図 1

$Van\ der\ Pole's\ equation : \dfrac{d^2y}{dx^2} + \mu(y^2 - 1)\dfrac{dy}{dx} + y = 0$

1 階連立微分方程式に書き直す。

$$\begin{cases} \dfrac{dy}{dx} = z \\ \dfrac{dz}{dx} = \mu(1 - y^2)z - y \end{cases}$$

初期値 $x = 0, y = 1, \dfrac{dy}{dx} = 0(= z), h = 0.1$

2.3 Script

```
NB. Differential Equation ==================================
NB. written by Toshio Nishikawa
NB. slightly modified by M.Shimura (2009/Feb.16)
NB. 0.1 (u=f) rk^:(i.11) 0 0

NB. Euler Adverb Defiition
NB. 0.1 ((1&{)-(0&{)) euler0^:(i.11) 0 0
euler0 =: 1 : 0
:
h0 =: x NB.  0.01
'X0 Y0' =:  y
k0 =: u X0, Y0
2{.(X0+h0), (Y0 + h0*k0)
)



NB.  Heun 法 ... 0.1 heun^:(i.11) 0 0
heun =: 1 : 0
:
h0 =. x
'X0 Y0' =:  y
k0 =.  u X0,Y0
k1 =.Y0 + h0*k0  NB. Euler
2{. (X0+h0),Y0+ (-: h0)* k0 + u (X0+h0),k1
NB. 2{. (X0+h0), (Y0 + h0*(k1 + k2)%2)
)

NB.  Revised Euler 法 ... 0.1 eulerx^:(i.11) 0 0
eulerx =: 1 : 0
:
h0 =. x
'X0 Y0' =:  y
k0 =: u X0,Y0
k1 =: Y0 + (-:h0) * k0
```

```
2{. (X0+h0), Y0 + h0* u (X0 + -:h0),k1
)

rk =: 1 : 0
:
h0 =. x
'X0 Y0' =:  y
k1 =:  u (X0),(Y0)
k2 =:  u (X0 + h0%2),(Y0 + h0*k1%2)
k3 =:  u (X0 + h0%2),(Y0 + h0*k2%2)
k4 =:  u (X0 + h0),(Y0 + h0*k3)
(X0+h0), (Y0 + h0*(k1 + (2*k2) + (2*k3) + k4)%6)
)

NB. ----------------------
f0=: 3 : '1-2 * y'   NB. 1-2x NB. Satoh/Nakamura Ex. 7.1
f1=: 3 : ' (0{ y)+1{ y' NB. x+y
f2=: 3 : '-(0{y)+1{y'

f3=: 3 : ' (3 * 1x1 ^ {.y)- ^&2 {:y' NB. Hattori Ex4.1


NB. 2nd order differential equation
rk_2nd =: 1 : 0
NB. Usage:  0.2 f10 rk_2nd ^:(i.6) 0 1 0
 :
hh=. -: h0 =. x
'X0 Y0 Z0'  =:  y
'K1 L1' =:  u L:0 X0,Y0,Z0
'K2 L2'  =: u  L:0 (X0 + hh),(Y0 + hh*K1),Z0+hh*L1
'K3 L3'  =: u  L:0 (X0 + hh),(Y0 + hh*K2),Z0+hh*L2
'K4 L4' =:  u  L:0 (X0 + h0),(Y0 + h0*K3),Z0+h0*L3
YY=. (Y0 + (h0 * +/ K1,(+: K2,K3),K4)%6)
ZZ=. (Z0 + (h0 * +/ L1,(+: L2,L3),L4)%6)
(X0+h0),YY,ZZ
)
```

```
NB. Satoh Nakamura
NB. Ex7.7
f10=: 3 : '({: y); _4* +/ }. y'
NB. dy/dx=z
NB. dz/dx=-4(y+z)
NB. EX7.8
f11=: 3 : '({: y);  (_8* 1{ y)-{:y'
NB. dy/dx=z
NB. dz/dx=-8y-z
NB. Ex Van der Pole equation
f12=: 4 : '({: y); ((1- (^&2) 1{ y)*x* {: y)- 1{y'
NB. dy/dx=z
NB. dz/dx=mu(1-y^2)*z-y NB. f12 is mu = 0
```

3 References

服部雄一 [Fortran による数値計算」1992 森北出版
石村貞夫 · 石村園子「ブラックショールズ微分方程式] 1999 東京図書
三井 · 小藤 · 斉藤 [微分方程式による計算科学入門] 2004 共立出版

付録 A　複数の関数と副詞

確率微分は半数が 2 個出てくる。これを副詞で裁く方法を整理する。

test0　複数の関数を同時に放り込む。同時に動作するが $u0, u1$ と切り分けできない。

```
test=: 1 : 0
'A B' =. y
A u B
)

t0=: 4 : 'x + y'
t1=: 4 : 'x * y'

   (t0;t1) test 1 2 3;4 5 6
+-----+-------+
|5 7 9|4 10 18|
+-----+-------+
```